```
> 
```

Partial Fractions, and Integration by Parts

```
> restart;
```

In this worksheet, we show how to explicitly implement integration by parts, and how to convert a proper or improper rational fraction to an expression with partial fractions.

Partial Fractions

We consider the function $f(x)=\dfrac{2x^5-8x^4+15x^3-10x^2-9x+27}{x^4-4x^3+5x^2-4x+4}$.

```
> f:=x -> (2*x^5-8*x^4+15*x^3-10*x^2-9*x+27)/(x^4-4*x^3+5*x^2-4*x+4);
```

$$f := x \to \frac{2x^5-8x^4+15x^3-10x^2-9x+27}{x^4-4x^3+5x^2-4x+4}$$

We use Maple to find its integral.

```
> Int(f(x),x)=int(f(x),x);
```

$$\int \frac{2x^5-8x^4+15x^3-10x^2-9x+27}{x^4-4x^3+5x^2-4x+4}\,dx = x^2-\frac{5}{x-2}+3\ln(x-2)+\ln(x^2+1)+7\arctan(x)$$

We now wish to use Maple to convert the integrand above by the method of partial fractions. When doing partial fractions by hand, we use the method only with proper fractions. Here, we use the convert command, where the argument parfrac refers to partial fraction format. This will work with improper fractions also.

```
> pf:=convert(f(x),parfrac,x);
```

$$pf := 2x+\frac{5}{(x-2)^2}+\frac{3}{x-2}+\frac{7+2x}{x^2+1}$$

```
> Int(pf,x)=int(pf,x);
```

$$\int 2x+\frac{5}{(x-2)^2}+\frac{3}{x-2}+\frac{7+2x}{x^2+1}\,dx = x^2-\frac{5}{x-2}+3\ln(x-2)+\ln(x^2+1)+7\arctan(x)$$

This is the same answer that we got above. To illustrate further, let us create separate expressions for the numerator and denominator of $f(x)$. We will use the numer and denom commands to do this.

```
> numerator:=numer(f(x));
```

$$numerator := 2x^5-8x^4+15x^3-10x^2-9x+27$$

```
> denominator:=denom(f(x));
```

$$denominator := x^4-4x^3+5x^2-4x+4$$

We form the original improper rational fraction.

```
> original:=numerator/denominator;
```

$$original := \frac{2x^5-8x^4+15x^3-10x^2-9x+27}{x^4-4x^3+5x^2-4x+4}$$

Since we have an improper rational fraction and the method of partial fractions is for proper rational fractions (degree of numerator less than degree of denominator), we use quo and rem to get the quotient and remainder from a long division of polynomials. Notice that the arguments for each command are "dividend, divisor, variable."

```
> quotient:=quo(numerator,denominator,x);
```

$$quotient := 2x$$

```
> remainder:=rem(numerator,denominator,x);
```

$$remainder := 27 + 5\,x^3 - 2\,x^2 - 17\,x$$

We form a rational fraction by dividing the remainder by the divsor.

```
> rf:=remainder/denominator;
```

$$rf := \frac{27 + 5\,x^3 - 2\,x^2 - 17\,x}{x^4 - 4\,x^3 + 5\,x^2 - 4\,x + 4}$$

We rewrite the original improper fraction as the sum of the quotient and a proper fraction.

```
> new:=quotient+rf;
```

$$new := 2\,x + \frac{27 + 5\,x^3 - 2\,x^2 - 17\,x}{x^4 - 4\,x^3 + 5\,x^2 - 4\,x + 4}$$

Again, we can use the convert command to convert the proper rational expression to partial fractions.

```
> rf:=convert(rf,parfrac,x);
```

$$rf := 5\,\frac{1}{(x-2)^2} + \frac{3}{x-2} + \frac{7 + 2\,x}{x^2 + 1}$$

The entire integrand is the sum of the quotient and the partial fraction decomposition of the proper fraction.

```
> integrand:=quotient+rf;
```

$$integrand := 2\,x + \frac{5}{(x-2)^2} + \frac{3}{x-2} + \frac{7 + 2\,x}{x^2 + 1}$$

Integration by Parts

Maple has a student package which is designed to illustrate calculus concepts in a step by step manner. We load this package by using the with statement.

```
> with(student);
Warning, the name integrand has been redefined
```

[D, *Diff*, *Doubleint*, *Int*, *Limit*, *Lineint*, *Product*, *Sum*, *Tripleint*, *changevar*, *completesquare*, *distance*, *equate*, *integrand*, *intercept*, *intparts*, *leftbox*, *leftsum*, *makeproc*, *middlebox*, *middlesum*, *midpoint*, *powsubs*, *rightbox*, *rightsum*, *showtangent*, *simpson*, *slope*, *summand*, *trapezoid*]

A list is given of all the new commands added. Our interest is in intparts. Let us apply this to $\int x\,\mathbf{e}^{(5\,x)}\,dx$. The intparts command takes two arguments. The first in the inert intergal we are interested in, and the second is the *u* from $\int u\,dv$.

```
> Int(x*exp(5*x),x)=intparts(Int(x*exp(5*x),x),x);
```

$$\int x\,\mathbf{e}^{(5\,x)}\,dx = \frac{1}{5}\,x\,\mathbf{e}^{(5\,x)} - \int \frac{1}{5}\,\mathbf{e}^{(5\,x)}\,dx$$

The command intparts can also be used with definite integration.

```
> Int(x*exp(5*x),x=2..4)=intparts(Int(x*exp(5*x),x=2..4),x);
```

$$\int_2^4 x\,\mathbf{e}^{(5\,x)}\,dx = \frac{4}{5}\,\mathbf{e}^{20} - \frac{2}{5}\,\mathbf{e}^{10} - \int_2^4 \frac{1}{5}\,\mathbf{e}^{(5\,x)}\,dx$$

Change of Variable

We can also use changevar for a change of variable. For example, suppose we wish to use the substitution $u = 2\,x$ in the integral $\int_a^b \frac{1}{\sqrt{1-4\,x^2}}\,dx$.

```
> Int(1/sqrt(1-4*x^2),x=a..b)=changevar(u=2*x,Int(1/sqrt(1-4*x^2),x=a..
  b),u);
```

$$\int_a^b \frac{1}{\sqrt{1-4\,x^2}}\,dx = \int_{2\,a}^{2\,b} \frac{1}{2}\,\frac{1}{\sqrt{1-u^2}}\,du$$

```
>
```